
Subject: Re: wait loop
Posted by David Fanning on Thu, 20 Sep 2001 13:51:07 GMT
View Forum Message <> Reply to Message

Ronn Kling (ronn@rlkling.com) writes:

>  What you want to use is a timer event for widgets.  This doesn't use any cpu
>  time until the event fires off.  If you don't have any widgets as part of
>  your program you can still create a simple base and keep map=0 so that it
>  never becomes visible.

Because I wasn't sure if an unmapped widget could
actually receive a timer event, I just coded a file-
seeking program up. Here it is. Note no error 
checking!

************************************************
PRO TestForFile_Event, event
Widget_Control, event.top, Get_UValue=filename
dummy = Findfile(filename, Count=count)
IF count EQ 0 THEN BEGIN
   Widget_Control, event.top, Timer=5 
   Print, 'Still looking...
ENDIF ELSE Print, 'Found File: ', filename
END

PRO TestForFile, filename
tlb = Widget_Base(Map=0)
Widget_Control, tlb, /Realize, Set_UValue=filename
Widget_Control, tlb, Timer=5
XManager, 'testforfile', tlb, /No_Block
END
**************************************************

I ran it like this:

   IDL> TestForFile, 'butter.pro'

Here is my output. Note, I had to save a file
named "butter.pro" to get the think stopped. :-)

   Still looking...
   Still looking...
   Still looking...
   Still looking...
   Found File: butter.pro

Cheers,

David

-- 
David W. Fanning, Ph.D.
Fanning Software Consulting
Phone: 970-221-0438, E-mail: david@dfanning.com
Coyote's Guide to IDL Programming: http://www.dfanning.com/
Toll-Free IDL Book Orders: 1-888-461-0155

